

## NAG Library Routine Document

### F07GSF (ZPPTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F07GSF (ZPPTRS) solves a complex Hermitian positive definite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where  $A$  has been factorized by F07GRF (ZPPTRF), using packed storage.

#### 2 Specification

SUBROUTINE F07GSF (UPLO, N, NRHS, AP, B, LDB, INFO)

INTEGER N, NRHS, LDB, INFO

COMPLEX (KIND=nag\_wp) AP(\*), B(LDB,\*)

CHARACTER(1) UPLO

The routine may be called by its LAPACK name *zpptrs*.

#### 3 Description

F07GSF (ZPPTRS) is used to solve a complex Hermitian positive definite system of linear equations  $AX = B$ , the routine must be preceded by a call to F07GRF (ZPPTRF) which computes the Cholesky factorization of  $A$ , using packed storage. The solution  $X$  is computed by forward and backward substitution.

If UPLO = 'U',  $A = U^H U$ , where  $U$  is upper triangular; the solution  $X$  is computed by solving  $U^H Y = B$  and then  $UX = Y$ .

If UPLO = 'L',  $A = LL^H$ , where  $L$  is lower triangular; the solution  $X$  is computed by solving  $LY = B$  and then  $L^H X = Y$ .

#### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

#### 5 Arguments

1: UPLO – CHARACTER(1) *Input*

*On entry:* specifies how  $A$  has been factorized.

UPLO = 'U'

$A = U^H U$ , where  $U$  is upper triangular.

UPLO = 'L'

$A = LL^H$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides.  
*Constraint:*  $NRHS \geq 0$ .
- 4: AP(\*) – COMPLEX (KIND=nag\_wp) array *Input*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the Cholesky factor of  $A$  stored in packed form, as returned by F07GRF (ZPPTRF).
- 5: B(LDB, \*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, NRHS)$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* the  $n$  by  $r$  solution matrix  $X$ .
- 6: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07GSF (ZPPTRS) is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 7: INFO – INTEGER *Output*  
*On exit:*  $INFO = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

$INFO < 0$

If  $INFO = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

For each right-hand side vector  $b$ , the computed solution  $x$  is the exact solution of a perturbed system of equations  $(A + E)x = b$ , where

$$\text{if UPLO} = 'U', |E| \leq c(n)\epsilon|U^H||U|;$$

$$\text{if UPLO} = 'L', |E| \leq c(n)\epsilon|L||L^H|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

If  $\hat{x}$  is the true solution, then the computed solution  $x$  satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(n) \text{cond}(A, x)\epsilon$$

where  $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_\infty / \|x\|_\infty \leq \text{cond}(A) = \| |A^{-1}| |A| \|_\infty \leq \kappa_\infty(A)$ .

Note that  $\text{cond}(A, x)$  can be much smaller than  $\text{cond}(A)$ .

Forward and backward error bounds can be computed by calling F07GVF (ZPPRFS), and an estimate for  $\kappa_\infty(A)$  ( $= \kappa_1(A)$ ) can be obtained by calling F07GUF (ZPPCON).

## 8 Parallelism and Performance

F07GSF (ZPPTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07GSF (ZPPTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $8n^2r$ .

This routine may be followed by a call to F07GVF (ZPPRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07GEF (DPPTRS).

## 10 Example

This example solves the system of equations  $AX = B$ , where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 3.93 - 6.14i & 1.48 + 6.58i \\ 6.17 + 9.42i & 4.65 - 4.75i \\ -7.17 - 21.83i & -4.91 + 2.29i \\ 1.99 - 14.38i & 7.64 - 10.79i \end{pmatrix}.$$

Here  $A$  is Hermitian positive definite, stored in packed form, and must first be factorized by F07GRF (ZPPTRF).

### 10.1 Program Text

```

Program f07gsfe

!      F07GSF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zpptrf, zpptrs
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, ldb, n, nrhs
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), b(:, :)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F07GSF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs

```

```

      ldb = n
      Allocate (ap(n*(n+1)/2),b(ldb,nrhs))

!      Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Factorize A
!      The NAG name equivalent of zpptrf is f07grf
      Call zpptrf(uplo,n,ap,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!          Compute solution
!          The NAG name equivalent of zpptrs is f07gsf
          Call zpptrs(uplo,n,nrhs,ap,b,ldb,info)

!          Print solution

!          ifail: behaviour on error exit
!          =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
          ifail = 0
          Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',
                    'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)
!
          Else
            Write (nout,*) 'A is not positive definite'
          End If

      End Program f07gsfe

```

## 10.2 Program Data

F07GSF Example Program Data

```

  4  2                                     :Values of N and NRHS
  'L'                                     :Value of UPLO
(3.23, 0.00)
(1.51, 1.92) ( 3.58, 0.00)
(1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
(0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00) :End of matrix A
( 3.93, -6.14) ( 1.48,  6.58)
( 6.17,  9.42) ( 4.65, -4.75)
(-7.17,-21.83) (-4.91,  2.29)
( 1.99,-14.38) ( 7.64,-10.79)                :End of matrix B

```

## 10.3 Program Results

F07GSF Example Program Results

```

Solution(s)
           1           2
1 ( 1.0000,-1.0000) (-1.0000, 2.0000)
2 (-0.0000, 3.0000) ( 3.0000,-4.0000)
3 (-4.0000,-5.0000) (-2.0000, 3.0000)
4 ( 2.0000, 1.0000) ( 4.0000,-5.0000)

```

---